Library of Congress Cataloging-in-Publication Data

Tuinenga, Paul W.
SPICE: a guide to circuit simulation and analysis using PSpice / Paul W. Tuinenga.–2nd ed.
p. cm.
Includes index,
ISBN 0-13-747270-6
1. SPICE (Computer program) 2. Electric circuit analysis–Data processing. I. Title.
TK454.T85 1992
621.381'5'02855369–dc20 91-34515
CIP

Acquisitions editor: ***Elizabeth Kaster***
Production editor: ***Irwin Zucker***
Copy editor: ***Brenda Melissaratos***
Cover design: ***Lundgren Graphics***

Prepress buyer: ***Linda Behrens***
Manufacturing buyer: ***David Dickey***
Supplements editor: ***Alice Dworkin***
Editorial assistant: ***Jaime Zampino***

TRADEMARK INFORMATION

IBM®PC is a registered trademark of International Business Machines Corporation.
PSpice® is a registered trademark of MicroSim Corporation.

Printed in the United States of America

10 9 8 7 6 5 4 3 2

ISBN 0-13-747270-6

Prentice-Hall International (UK) Limited, *London*
Prentice-Hall of Australia Pty. Limited, *Sydney*
Prentice-Hall Canada Inc., *Toronto*
Prentice-Hall Hispanoamericana, S.A., *Mexico*
Prentice-Hall of India Private Limited, *New Delhi*
Prentice-Hall of Japan, Inc., *Tokyo*
Simon & Schuster Asia Pte. Ltd., *Singapore*
Editora Prentice-Hall do Brasil, Ltda., *Rio de Janeiro*

CHAPTER 8

LOOP GAIN ANALYSIS

Determining the loop gain of circuits with high-gain components is a difficult task. You might be tempted to open the loop to make measurements, but this will probably destroy the DC bias of the circuit. Opening the loop might also disconnect an internal load impedance and affect your measurements. What we need is a way to make these measurements **without opening the loop**† or changing the internal loading of the circuit.

In developing this technique we will start with a hypothetical circuit. We will set up this circuit to make these measurements without introducing any problems. Then we will modify the formulas to work with "real" circuits. But first, let us review some terms.

8.1 AN IDEAL CIRCUIT



FIGURE 8-1 Schematic of system with feedback.

In a system with feedback, shown in Figure 8-1, the following signals may be calculated by inspection:

$$\begin{aligned} u_o &= A \cdot u_e \\ u_e &= u_i - K \cdot u_o \end{aligned} \tag{8-1}$$

† This technique, as far as the author can determine, originates from a *Hewlett-Packard Applications Note*, circa 1965. The technique has been extended and refined by Dr. R. D. Middlebrook of the California Institute of Technology; see *International Journal of Electronics*, vol. 38, no. 4 (1975), 485-512.

where u_i, u_o, and u_e are the *input*, *output*, and *error* signals, respectively. Further manipulation of these formulas yields

$$u_o = A \cdot u_i - A \cdot K \cdot u_o$$

$$u_e = \frac{u_i}{1 + A \cdot K} \tag{8-2}$$

so that system gain, G, is

$$G = \frac{u_o}{u_i} = \frac{A}{1 + A \cdot K} = \frac{A}{1+T} = \frac{1}{K}\frac{T}{1+T} \tag{8-3}$$

where the *loop gain*, or *return ratio*, T is

$$T = \frac{u_f}{u_e} = A \cdot K \tag{8-4}$$

The system's relative stability can be inspected in an open-loop configuration. The loop is opened in the feedback path, and a "test" signal is injected. Then the resulting feedback signal, opposite the injection point, is compared to the test signal. The feedback signal is inspected for one full cycle (360°) of phase shift. However, since the feedback is subtracted from the input, the subtraction alone provides 180° of phase shift. So the feedback signal should be inspected for an additional 180° of phase shift, and not 360°. If the loop gain (the ratio of feedback signal to the test signal) is one, or greater, the loop is unstable, since it can supply its own input. The amount of gain, relative to unity, at 180° phase shift is called the loop *gain margin*. Likewise, the amount of phase difference from 180°, when the loop gain is unity, is called the loop *phase margin*.

Actually this analysis is true even if the loop is broken in the forward path, but it seems easier to describe as though the feedback path were broken. Now, having said that, we will imagine that we will "break" the loop somewhere inside the circuit. It doesn't matter where, just as long as the break is in some part of the signal path. To do this, we will imagine that some part of the signal path is a controlled-current source connected to an impedance (for example, imagine an ideal transistor with a load resistance) as shown in Figure 8-2. Also, the normal input signal will be set to zero so that we need only consider the effects of the test signal.



FIGURE 8-2 Idealized section of loop circuitry.



FIGURE 8-3 Load duplicated, current injected.



FIGURE 8-4 Via superposition, loop is not broken.



FIGURE 8-5 Voltage-mode equivalent.

Then, we break the loop, as shown in Figure 8-3, being careful to duplicate the load impedance seen by the current source. Now we inject a current into the original load impedance, which is part of the original loop, as the test signal. The return signal is current in the duplicated load impedance. The ratio of the two signals is merely the **current loop gain** measurement, or Ti, where Ti = i_y/i_x.

Now comes part of the trick: it is not necessary to open the loop to inject the test signal. As shown in Figure 8-4, if a current is injected directly into the signal path, it splits into the test signal and return signal of Figure 8-3. Furthermore, the load impedance does not need to be duplicated, since the loop is not broken and the current source has infinite impedance.

Similarly, we can develop the same technique using voltages instead of currents. As shown in Figure 8-5, a voltage source is inserted in the loop to inject a signal. The resulting measurement of the voltage across the load impedance and the controlled-voltage source yields a ratio that is merely the **voltage loop gain** measurement, or Tv, where Tv = v_y/v_x.

Since we were able to choose ideal points to break the loop, the signal ratios are the loop gain of the system; that is, T ≡ Ti for the current measurement, and T ≡ Tv for the voltage measurement.

8.2 A "REAL" CIRCUIT

Now we tackle a "real" circuit. Our example so far assumed controlled sources that have no internal impedance the way a real transistor, or real opamp, does. We can account for this impedance by use of superposition.



FIGURE 8-6 Ideal injection.



FIGURE 8-7 Real injection.

In the previous case the ideal injection was performed as shown in Figure 8-6, but in this case we have two new currents, as shown in Figure 8-7:

$$i'_x = \frac{Z_2}{Z_1+Z_2} i_x = \frac{i_x}{1+\frac{Z_1}{Z_2}} \tag{8-5}$$

$$i'_y = i_y + \frac{Z_1}{Z_1+Z_2} i_x = M \cdot i'_x + \frac{Z_1}{Z_2} i'_x$$

where M is the rest of the loop's gain. The current loop ratio that we measure, then, is

$$Ti = \frac{i'_y}{i'_x} = M + \frac{Z_1}{Z_2} \tag{8-6}$$

and the loop gain is

$$T = \frac{i_y}{i_x} = \frac{M \cdot i'_x}{\left(1+\frac{Z_1}{Z_2}\right) i'_x} = \frac{M+\frac{Z_1}{Z_2}-\frac{Z_1}{Z_2}}{1+\frac{Z_1}{Z_2}} = \frac{Ti-\frac{Z_1}{Z_2}}{1+\frac{Z_1}{Z_2}} \tag{8-7}$$

Notice that this is the measurement of a real circuit, where the active element has been replaced with its Norton-equivalent current source and impedance, Z_2. The remaining impedance, Z_1, represents the load of the active element.

By similar means, we might make a non-ideal measurement of the voltage loop ratio, as shown in Figure 8-8, with the result

$$T = \frac{Tv-\frac{Z_2}{Z_1}}{1+\frac{Z_2}{Z_1}} \tag{8-8}$$

Again, notice that this is the measurement of a real circuit, where the active element has been replaced with its Thévinin-equivalent voltage source and impedance, Z_2. The remaining impedance, Z_1, represents the load of the active element.

Of course, this begs the question of what these circuit impedances are so we may calculate T exactly. By inspection, we can see that

$$T \approx Ti, \quad \text{if } \frac{Z_1}{Z_2} \ll 1 \text{ and } \frac{Z_1}{Z_2} \ll T$$

$$T \approx Tv, \quad \text{if } \frac{Z_2}{Z_1} \ll 1 \text{ and } \frac{Z_2}{Z_1} \ll T \tag{8-9}$$

Notice that if we were to measure both Ti and Tv, we would have two equations with two unknowns. To eliminate the impedance ratio, we first rewrite the measurement



FIGURE 8-8 Real injection for voltage measurement.

ratios in terms of T

$$\mathrm{Ti} = \left(1+\frac{Z_1}{Z_2}\right)\mathrm{T} + \frac{Z_1}{Z_2} \qquad \text{(8-10)}$$

$$\mathrm{Tv} = \left(1+\frac{Z_2}{Z_1}\right)\mathrm{T} + \frac{Z_2}{Z_1}$$

and then, by adding 1 to both sides and adding the reciprocals, we find that

$$(\mathrm{T}+1) = (\mathrm{Ti}+1)\,||\,(\mathrm{Tv}+1) \qquad \text{(8-11)}$$

where || means "parallel combination"; for example:

$$x\,||\,y = \frac{1}{\frac{1}{x}+\frac{1}{y}} = \frac{x\,y}{x+y} \qquad \text{(8-12)}$$

This says that the lower of the two measurements, Ti or Tv, dominates the value of T, the loop gain. It also says that we can make both measurements, as shown in Figure 8-9, and calculate T exactly. Another way to restate the formula for T is

$$\mathrm{T} = \frac{\mathrm{Ti}\cdot\mathrm{Tv}-1}{\mathrm{Ti}+\mathrm{Tv}+2} \qquad \text{(8-13)}$$

which may be more suitable for numerically stable calculations.



FIGURE 8-9 Making Ti and Tv measurements.

8.3 A "REAL" EXAMPLE

Let's try to calculate loop gain in a relatively simple circuit. Figure 8-10 shows the circuit we will measure, breaking into the loop at the output of the opamp.

The simplified model of the opamp (also shown) will be used for this example. The subcircuit definition for the opamp is

```
* "ideal" op-amp with 100K gain and one-pole roll-off at 10Hz
.subckt opamp non inv out
  rin non inv 100K
  egain 1 0 (non,inv) 100K
  ropen 1 2 1K
  copen 2 0 15.92u
  eout  3 0 (2,0) 1
  rout  3 out 50
.ends
```

Since we will want two copies of the entire circuit we are measuring, let's put the circuit in a subcircuit. This subcircuit will have only two nodes, which are at the place where we are breaking the loop:

```
* example circuit
.subckt test left right
  vin 1 0 DC 0
  x1 1 2 left opamp
  r1 right 0 200
  r2 right 2 10K
  r3 2 0 1K
  c1 2 0 .038u
.ends
```

Finally, there is the rest of the circuit:

```
* Loop gain measurement
.ac dec 100 1 1Meg
.probe
xi Ti_left Ti_right test ; this copy for Ti measurement
xv Tv_left Tv_right test ; this copy for Tv measurement

* perform Ti measurements
iz 0 1 AC 1 ; current stimulus
viy 1 Ti_left  DC 0 ; sense Ix
hiy iy 0 viy 1      ; convert Ix to a voltage
riy iy 0 1G
vix 1 Ti_right DC 0 ; sense Iy
hix ix 0 vix 1      ; convert Iy to a voltage
rix ix 0 1G

* perform Tv measurements
vz Tv_right Tv_left AC 1 ; voltage stimulus
evy vy 0 (0,Tv_left ) 1 ; duplicate Vx
rvy vy 0 1G
evx vx 0 (Tv_right,0) 1 ; duplicate Vy
rvx vx 0 1G
```

Notice that we may break into the loop of this example circuit elsewhere, or try another circuit, just by changing the description of the test subcircuit.



FIGURE 8-10 Loop gain measurement example circuit.

After running an AC analysis simulation, we view the results with the aid of the following macro definitions. (Probe macro definition and use are covered in §6.6.)

```
par(a,b)=(((a)*(b))/((a)+(b)))
Ti=(V(iy)/V(ix))
Tv=(V(vy)/V(vx))
T=(par(Ti+1,Tv+1)-1)
```

The first macro, par(a,b), defines the "parallel" operation (that is, $a\,||\,b$) and is written to be numerically stable as the arguments approach zero. Without macros, or Probe handling complex arithmetic, displaying the loop functions becomes quite an ordeal. For example, the relatively simple expression for the magnitude of Ti+1 is

$$|Ti+1| = \sqrt{\frac{\left(ix_{re} - iy_{re}\right)^2 + \left(ix_{im} - iy_{im}\right)^2}{ix_{re}^2 + iy_{im}^2}} \tag{8-14}$$

where, for example, ix_{re} is the real part of ix.

First we look at the magnitude response of T, Ti, and Tv, as shown in Figure 8-11. As we would expect, at lower frequencies Tv dominates, since the input impedance of the feedback circuit is much greater than the output impedance of the opamp; that is, Tv is of smaller magnitude and will control the value of (Ti+1)||(Tv+1). But as the frequency increases, the impedance of the feedback circuit decreases and begins to load the opamp. As the loop gain components, Ti and Tv, approach unity, or 0dB, their contribution is dominated by the +1 in the calculation of T derived in (8-11). Weird and non-intuitive things happen to Ti and Tv as frequency increases, but T has the shape we expect from knowing the frequency response of the opamp and the feedback circuit.

Now we look at the phase response of T, Ti, and Tv, as shown in Figure 8-12. Here we find that, at low frequencies, the phase is controlled by the opamp. But



FIGURE 8-11 Plot of open-loop magnitude responses.

again, as frequency increases, the current and voltage components, Ti and Tv, give an inaccurate indication of the loop phase response. Intuitively, we know that shape of T is correct from knowing the opamp's phase response, which has a single pole, and that the feedback circuit has no resonant circuitry that would give the 180° phase shift of a double pole.



FIGURE 8-12 Plot of open-loop phase responses.

8.4 UNSTABLE LOOP GAIN

Every once in a while, through no fault of the engineer (of course), a system is designed that is unstable or only conditionally stable. The problem then becomes how to correct this, but the type and amount of correction needed depend on how "bad" the system is. How do we ascertain this?

The key to measuring an unstable loop comes from reconsidering the Ti and Tv measurements we made earlier. As shown in Figure 8-7 and Figure 8-8, in each case two signals are created and measured. The ratio of the signals is some type of loop gain measurement. Analogues of those figures are shown in Figure 8-13 and Figure 8-14, where now we assume the generating sources injecting the test signal have finite impedance. However, we can easily see that this does not affect the calculation of Ti or Tv. In both cases, the ratio of the current or voltage values, now *primed* values in the newer figures, will be the same as before.

But the circuit has changed! The impedance of the test signal source changes the loading on the active device, yet the loop gain measurement is not changed. This happens because the effect of the source impedance is not part of the measured values. Another way of saying this is that the measurements have been designed to be taken outside of the subnetwork that contains the source impedance.

But the source impedance still loads the circuit! This gives us a mechanism for altering the circuit's response without affecting the original loop gain measurement. In particular, we can use the source impedance, or an additional impedance associated with the source, to lower the loop gain so that it becomes marginally stable. Then we make our measurements to calculate the loop gain.

In practice, forcing the loop to be stable is only a concern for real circuits. Before starting small-signal analysis, the simulator calculates an operating point ignoring all capacitances and inductances, which are the effects that usually create instability. PSpice allows you to simulate and measure an unstable loop gain without resorting to a loading impedance to stabilize the loop, usually. However, you might come across a circuit that is DC unstable, having net positive feedback at zero frequency. In this case, PSpice will have problems finding an operating point and you need to use the technique just discussed to analyze such a circuit.



FIGURE 8-13 Current-mode measurement.



FIGURE 8-14 Voltage-mode measurement.